
 The data from two watches – deviances from exact time

In order to look at the performance of two different ordinary dial-watches we com​pared the reading against the time presented in the upper right hand corner of our TV. To make it easier we took the readings when the TV showed a full minute. The whole procedure was per​formed by one person and the result is recorded as integer seconds. In a proper experiment the reading should probably be taken at equal time intervals, this was not the case here as can be seen by the plots below. The watches were not set or changed in any way before the experi​ments. See also Appendix – some useful menus fore support by the software.
Initial summary.  The graph shows the deviations (seconds, on Y-axis) from the two watches measured over a number of weeks. There is an obvious drift in both the watches and at different slopes. This is called an interaction between the variable 'watch' and 'time'. (Last page contains a photo of the wathes involved.)
[image: image1.png]All data

12

10

03-06-21

03-07-01

All deviations for the two watches
The two series starts from zero

03-07-11 030721
True time

03-08-01




Below there is data from an old clock and it is obvious by looking at the data that the old clock has a dramatically larger drift from correct value.
Entering of data.  We start by entering the data from the TV and then the readings. A special column below, c3, will be used to designate ‘Watch 1’ and ‘Watch 2’.

erase c1-c100                      # Erases the first 100 columns.

name c1 'TV-time' c2 'Recorded time' c3 'Watch'

name c4 'Diff in sec'

# Enters the time readings of the time on TV at measuring occasions of two

# different watches. The format is specified in 'FORMAT'.

read c1;

format(DT40yy-mm-dd hh:mm:ss).

03-06-20 10:32:00

03-06-20 10:32:00

03-06-22 21:05:00

03-06-22 21:06:00

03-07-01 21:48:00

03-07-01 21:46:00

03-07-22 22:44:00

03-07-22 22:43:00

03-07-31 21:39:00

03-07-31 21:40:00

03-08-03 21:50:00

03-08-03 21:51:00

03-08-05 22:07:00

03-08-05 22:08:00

end

read c2;                           # Enters the readings of two different

format(DT40yy-mm-dd hh:mm:ss).     # watches at times specified in c2.

03-06-20 10:31:57

03-06-20 10:35:51

03-06-22 21:04:58

03-06-22 21:09:52

03-07-01 21:48:00

03-07-01 21:49:55

03-07-22 22:44:03

03-07-22 22:47:00

03-07-31 21:39:04

03-07-31 21:44:02

03-08-03 21:50:05

03-08-03 21:55:02

03-08-05 22:07:06

03-08-05 22:12:03

end

set c3                             # Enters a '1' and a '2' to separate the

7(1 2)                             # readings.

end

let c4 = (c2 - c1)*24*60*60        # Calculates the difference between the TV and

                                   # the watches (result expressed in seconds).

Name c5 'True W1' c6 'Meas W1' c7 'Diff W1'

copy c1 c2 c4 c5 c6 c7;            # Copies 'true', 'measured', 'difference' for

use c3 1.                          # watch 1. Stores result in c5-c7.

Name c9 'True W2' c10 'Meas W2' c11 'Diff W2'

copy c1 c2 c4 c9 c10 c11;          # Copies 'true', 'measured', 'difference' for

use c3 2.                          # watch 2. Stores result in c10-c11.

plot c4*c1;                        # Plots all results.

title 'All deviations for the two watches'.

The preceding plot is not very useful. As there was a large difference between the settings of the two watches at the start of the experiment, most of the deviations or drifts of the watches are concealed by the drawn out scale of the Y-axis. Plotting the results individually will be better: 

plot c7*c5;                        # Plots result for watch 1.

title 'Deviations for watch 1'.
plot c11*c9;                       # Plots result for watch 2.

title 'Deviations for watch 2'.
If we subtract the first value for from all values in each dataseries we get two series that starts from zero and thus we can plot then in the same diagram. The following commands do this and create a new plot:

Name c13 'Data W1' c14 'Data W2' c15 'All data' c16 'True time'

let c13 =  c7 -  c7(1)             # Subtracts the first value (watch 1).

let c14 = c11 - c11(1)             # Subtracts the first value (watch 2).

stack c13 c14 c15                  # Stacks all deviations in c15 (Y).

stack  c5  c9 c16                  # Stack all measuring times (X).

plot c15*c16;                      # Plots all data.

title 'All deviations for the two watches';

title 'The two series starts from zero'.
	Below various regression analyses are performed but because of a change in commands the document has been split in two parts. In order to avoid confusion all texts are duplicated but edited where necessary.

· for use with Minitab version 17 or higer

· for use with Minitab version 16 or lower



Regression analysis I (Minitab 17 or higher).  As we know that the data consists of two diffe​rent watches it seems to be silly to fit a straight line to the data. Therefore we use a variable for the two watches. Thus the fitted model is the following where X1 is time when the reading was taken and X2 is watch number:

Let k1 = n(c13)                    # Counts number of values in c13 (W1).

Let k2 = n(c14)                    # Counts number of values in c14 (W2).

Name c17 'Clock' c22 'Pred'        # 'Clock'-variable and predicted values.

Set c17                            # -1 = watch 1, 1 = watch 2.

K1(-1) k2(1)

End

Set c21                            # 1 = watch 1 (black), 2 = watch 2 (red)

K1(1) k2(2)                        # used for colouring dots in the plot.

End

Regress;                           # Explains deviation using 'All data'.
  Response 'All data';

  Nodefault;

  Continuous 'True time';

  Categorical 'Clock';

  Terms 'True time' clock;

  Constant;

  Unstandardized;

  coeffiecents c20;                # Stores the coefficient in c20.
  Tanova;

  Tsummary;

  Tcoefficients;

  Tequation;

  TDiagnostics 0.

Predict 'All data';                # Used to get predicted values for the
  Nodefault;                       # plot below.
  CPredictors 'True time' Clock;

  PFITS Pred.

Copy c22 c16 c31 c32;              # Used to draw lines in the plot below.
  Include;

  Where "c21 = 1".
Copy c22 c16 c33 c34;              # Used to draw lines in the plot below.
  Include;

  Where "c21 = 2".
plot c15*c16;                      # Plots all data including estimated model.

title 'All deviations for the two watches';

tsize 1.1;

title 'The two series starts from zero';

tsize 0.9;

tcolor 4;

symbol;

color c21;

line c34 c33;                     # Draws a black line.
color 1;

line c32 c31;                     # Draws a red line.
color 2.

After seeing the plot and knowing the background of the technical situation we realize that the analysis should consist of two models showing different slopes. Therefore we perform a second regression analysis where we let the slope interact with the type of watch.
Regression analysis II (Minitab 17 or higher).  We introduce an interaction-term, X1X2, in the model:

Regress;                           # Explains deviation using 'All data'.
  Response 'All data';

  Nodefault;

  Continuous 'True time';

  Categorical 'Clock';

  Terms 'True time' clock 'True time'*clock;

  Constant;

  Unstandardized;

  coeffiecents c20;                # Stores the coefficient in c20.
  Tanova;

  Tsummary;

  Tcoefficients;

  Tequation;

  TDiagnostics 0.

Predict 'All data';                # Used to get predicted values for the
  Nodefault;                       # plot below.
  CPredictors 'True time' Clock;

  PFITS Pred.

Copy c22 c16 c31 c32;              # Used to draw lines in the plot below.
  Include;

  Where "c21 = 1".
Copy c22 c16 c33 c34;              # Used to draw lines in the plot below.
  Include;

  Where "c21 = 2".
plot c15*c16;                      # Plots all data including estimated model.

title 'All deviations for the two watches';

tsize 1.1;

title 'The two series starts from zero';

tsize 0.9;

tcolor 4;

symbol;

color c21;

line c34 c33;                     # Draws a black line.
color 1;

line c32 c31;                     # Draws a red line.
color 2.
This analysis produces a graph with two lines with different slopes.
A prediction (Minitab 17 or higher).  We also have another set of measurements three weeks after the final measure​ment above. How will these points fit? How will the two lines predict these values? We need to add these points and treat the data as above. (We suppose that the original data still is in the worksheet. Information about predicting using the REGRESSION-command can be found typing HELP REGRESSION):

read c22;                          # Two extra readings 'exact times' (TV).

format(DT40yy-mm-dd hh:mm:ss).     # 

03-08-26 22:24:00

03-08-26 22:25:00

end

read c23;                          # Two extra readings: watch 1 and watch 2.

format(DT40yy-mm-dd hh:mm:ss).     # 

03-08-26 22:24:11

03-08-26 22:29:05

end

let c24(1) = (c23(1)-c22(1))*24*60*60 -  c7(1) # Extra deviation watch 1.

let c24(2) = (c23(2)-c22(2))*24*60*60 - c11(1) # Extra deviation watch 2.

stack c15 c24 c15                  # Stack old deviations on new deviations.

stack c16 c22 c16                  # Old time of measure on new measures.

let c21(15) = 3                    # Colour code new values: green.

let c21(16) = 3
                                   # The two new points are not used for

                                   # any re-calculations.

plot c15*c16;                      # Plots all data incl late data.

title 'All deviations for the two watches';

tsize 1.1;

title 'Black: watch 1.  Red: watch 2';

tsize 0.9;

tcolor 4;

symbol;

color c21;

axlabel 2 'Deviations in seconds';

line c34 c33;                     # Draws a black line.
color 1;

line c32 c31;                     # Draws a red line.
color 2.

We see from the plot that a formal prediction of the deviations some three weeks later will coin​cide fairly well with the actual values obtained from the watches.

An old clock (Minitab 17 or higher).  We have also done the same measurement on another older clock. This investi​ga​tion, performed 2003, actually triggered the experiment above. We add the data separately and show a plot below and perform a simple regression analysis. This data could also have been in​clu​ded in the previous analysis (which of course would have needed an extended model). Here the 'true' value of time was obtained by using the telephone service:

erase c1-c100                      # Erases the worksheet.

read c1;                           # True time from the telephone service.

format(DT40yy-mm-dd hh:mm:ss).

02-06-05 19:36:00

02-06-06 09:24:20

02-06-06 19:41:10

02-06-07 08:15:40

02-06-07 22:43:30

02-06-08 20:21:20

02-06-09 11:57:30

02-06-10 09:34:40

02-06-10 22:44:40

02-06-11 11:02:40

02-06-12 22:39:10

end

read c2;                           # Enters the readings of the old clock.

format(DT40yy-mm-dd hh:mm:ss).

02-06-05 19:37:12

02-06-06 09:25:38

02-06-06 19:42:33

02-06-07 08:17:12

02-06-07 22:45:15

02-06-08 20:23:18

02-06-09 11:59:35

02-06-10 09:37:01

02-06-10 22:47:14

02-06-11 11:05:20

02-06-12 22:42:20

end

let c4 = (c2 - c1)*24*60*60        # Calculates the difference (old clock).

plot c4*c1;                        # Plots all results.

title 'All deviations for the old clock'.
fitline c4 c2                      # Analysis of data from old clock.
This time we got a slope of 16.5, which means a drift of, on the average, 16.5 seconds per day. This is approximately 100 times more than the best one of the two watches analysed above!

Regression analysis I (Minitab 16 or lower).  As we know that the data consists of two different watches it seems to be silly to fit a straight line to the data. Therefore we use an indicator variable for the two watches. In general it is sound to have indicator variables with an average of 0. Therefore we create a centred indicator variable before doing the regression analysis. Thus the fitted model is the following where X1 is time when the reading was taken and X2 is watch number:

Let k1 = n(c13)                    # Counts number of values in c13 (W1).

Let k2 = n(c14)                    # Counts number of values in c14 (W2).

Name c17 'indi'                    # Indicator variable.

Set c17                            # -1 = watch 1, 1 = watch 2.

K1(-1) k2(1)

End

Set c21                            # 1 = watch 1 (black), 2 = watch 2 (red)

K1(1) k2(2)                        # used for colouring dots in the plot.

End

Regression c15 2 c16 c17;          # Explains deviation using 'time' and

Coeff c20.                         # 'watch'. Coefficients are stored in c20.

Let k11 = min(c16)                 # Creates endpoints for lines in plot.

Let k12 = max(c16)

Let k13 = c20(1) + c20(2)*k11 + c20(3)*(-1)

Let k14 = c20(1) + c20(2)*k11 + c20(3)*( 1)

Let k15 = c20(1) + c20(2)*k12 + c20(3)*(-1)

Let k16 = c20(1) + c20(2)*k12 + c20(3)*( 1)

plot c15*c16;                      # Plots all data including estimated model.

title 'All deviations for the two watches';

tsize 1.1;

title 'The two series starts from zero';

tsize 0.9;

tcolor 4;

symbol;

color c21;

line k11 k13 k12 k15;

color 1;

line k11 k14 k12 k16;

color 2.

After seeing the plot and knowing the background of the technical situation we realize that the analysis should consist of two models showing different slopes. Therefore we perform a second regression analysis where we let the slope interact with the type of watch.
Regression analysis II (Minitab 16 or lower).  We introduce an interaction-term, X1X2, in the model:

let c18 = c16*c17                 # Creating interaction term (X1X2).

Regression c15 3 c16-c18;         # Explains deviations by three variables.

Coeff c20.                        # The four coefficients are stored in c20.

Let k11 = min(c16)

Let k12 = max(c16)

Let k13 = c20(1) + c20(2)*k11 + c20(3)*(-1) + c20(4)*k11*(-1)

Let k14 = c20(1) + c20(2)*k11 + c20(3)*( 1) + c20(4)*k11*( 1)

Let k15 = c20(1) + c20(2)*k12 + c20(3)*(-1) + c20(4)*k12*(-1)

Let k16 = c20(1) + c20(2)*k12 + c20(3)*( 1) + c20(4)*k12*( 1)

plot c15*c16;                      # Plots all data.

title 'All deviations for the two watches';

tsize 1.1;

title 'Black: watch 1.  Red: watch 2';

tsize 0.9;

tcolor 4;

symbol;

color c21;

axlabel 2 'Deviations in seconds';

line k11 k13 k12 k15;              # Line showing model for watch 1.

color 1;

line k11 k14 k12 k16;              # Line showing model for watch 2.

color 2.

Comments to regression analysis II (Minitab 16 or lower).  We are of course interested in the coefficients of the straight lines. These will be an estimate how much the watches drift per day. We enter the infor​mation in the estimated model where we show only those coefficients that are valid. The two cal​culations give us the two different slopes:

Let us see if these values are reasonable. The following command gives us information of the duration of the test:

let k100 = max(c16)-min(c16)       # Difference between smallest and largest

prin k100                          # value of the X-axis (approx 46.5 days).

let k100 = k15-k13                 # Difference between smallest and largest

prin k100                          # Y-value of black line (appr 8.0 seconds).

let k100 = k16-k14                 # Difference between smallest and largest

prin k100                          # Y-value of red line (approx 11.6 seconds).

From this we calculate 8/46.5 giving 0.17 and 11.6/46.5 giving 0.247 and thus we think that our use of the model seems to be reasonable. The first watch changes on the average 0.17 seconds per day and the second watch correspondingly 0.25 seconds per day.

A prediction (Minitab 16 or lower).  We also have another set of measurements three weeks after the final measure​ment above. How will these points fit? How will the two lines predict these values? We need to add these points and treat the data as above. (We suppose that the original data still is in the worksheet. Information about predicting using the REGRESSION-command can be found typing HELP REGRESSION):

read c22;                          # Two extra readings 'exact times' (TV).

format(DT40yy-mm-dd hh:mm:ss).     # 

03-08-26 22:24:00

03-08-26 22:25:00

end

read c23;                          # Two extra readings: watch 1 and watch 2.

format(DT40yy-mm-dd hh:mm:ss).     # 

03-08-26 22:24:11

03-08-26 22:29:05

end

let c24(1) = (c23(1)-c22(1))*24*60*60 -  c7(1) # Extra deviation watch 1.

let c24(2) = (c23(2)-c22(2))*24*60*60 - c11(1) # Extra deviation watch 2.

stack c15 c24 c15                  # Stack old deviations on new deviations.

stack c16 c22 c16                  # Old time of measure on new measures.

let c21(15) = 3                    # Colour code new values: green.

let c21(16) = 3                    # 

                                   # The two new points are not used for

                                   # any re-calculations.

plot c15*c16;                      # Plots all data incl late data.

title 'All deviations for the two watches';

tsize 1.1;

title 'Black: watch 1.  Red: watch 2';

tsize 0.9;

tcolor 4;

symbol;

color c21;

axlabel 2 'Deviations in seconds';

line k11 k13 k12 k15;

color 1;

line k11 k14 k12 k16;

color 2.

We see from the plot that a formal prediction of the deviations some three weeks later will coin​cide fairly well with the actual values obtained from the watches.

An old clock (Minitab 16 or lower).  We have also done the same measurement on another older clock. This investi​ga​tion, performed 2003, actually triggered the experiment above. We add the data separately and show a plot below and perform a simple regression analysis. This data could also have been in​clu​ded in the previous analysis (which of course would have needed an extended model). Here the 'true' value of time was obtained by using the telephone service:

erase c1-c100                      # Erases the worksheet.

read c1;                           # True time from the telephone service.

format(DT40yy-mm-dd hh:mm:ss).

02-06-05 19:36:00

02-06-06 09:24:20

02-06-06 19:41:10

02-06-07 08:15:40

02-06-07 22:43:30

02-06-08 20:21:20

02-06-09 11:57:30

02-06-10 09:34:40

02-06-10 22:44:40

02-06-11 11:02:40

02-06-12 22:39:10

end

read c2;                           # Enters the readings of the old clock.

format(DT40yy-mm-dd hh:mm:ss).

02-06-05 19:37:12

02-06-06 09:25:38

02-06-06 19:42:33

02-06-07 08:17:12

02-06-07 22:45:15

02-06-08 20:23:18

02-06-09 11:59:35

02-06-10 09:37:01

02-06-10 22:47:14

02-06-11 11:05:20

02-06-12 22:42:20

end

let c4 = (c2 - c1)*24*60*60        # Calculates the difference (old clock).

plot c4*c1;                        # Plots all results.

title 'All deviations for the old clock'.
regression c4 1 c2                 # Analysis of data from old clock.
This time we got a slope of 16.5, which means a drift of, on the average, 16.5 seconds per day. This is approximately 100 times more than the best one of the two watches analysed above!

Appendix – some useful menus

There are some useful menus when looking at time and/or converting or extracting time measurements.

	This menu is used for converting different types of data.

From or to date/text/numerical.

By right-clicking the Date/Time column, the very format of the column can be changed.

	This menu, together with the one below, is used to extract certain parts of the date/time data.

	Perhaps there is a need to plot or stratify the data per week day in order to investigate of visualize the variation.

By clicking e.g. 'Day of week' this can be achieved. The result will be either as text or as numerical values, de​pending of the choice of the previous menu above.


Some reflexions

	The photo of the watches (left) shows 'watch 1', 'watch 2' and 'an old clock'. Watch 1 was new at November 2002 while watch 2 and the old clock are some 15-20 years old. Watch 2 has been running all the time while the old clock needs to be rewound appro​ximately every day and thus is only in use when it is found in the drawer.

Watch 2 shows one hour difference from watch 1 and the rea​son is that it has not bee changed when changing winter/summer.

This experiment shows some features of experimenting. In general it is always good to run a short experiment and get 'hands on' experience of the equipment, the measurements, the re​cording of data.

Thus we need to be two people taking the true readings from TV: one person giving an audio indication like 'Now!' for the second person to read the watches. (It was difficult for a person staring at the TV and then shifting focus taking a correct reading of the watch. For one person it was easier to use the telephone service which gave an audio signal at every 10 seconds.)

	As we are getting the results in integer seconds, and thus intro​ducing rounding error, perhaps we should add some random component.

In the analysis above we had both watches in the same analysis. Why not doing two different ana​lyses, one for each watch? When we put everything into one analysis we get more so-called de​grees of freedom when we estimate the residual variation. This variation is used for impor​tant parts of the analysis and the more degrees of freedom, the better estimate. If we had split the analysis into different ones we would have got two different estimates, based on fewer de​grees of freedom and thus less accurate. ■

